

NAG Library Routine Document

C05QBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

C05QBF is an easy-to-use routine that finds a solution of a system of nonlinear equations by a modification of the Powell hybrid method.

2 Specification

```
SUBROUTINE C05QBF (FCN, N, X, FVEC, XTOL, IUSER, RUSER, IFAIL)
  INTEGER          N, IUSER(*), IFAIL
  REAL (KIND=nag_wp) X(N), FVEC(N), XTOL, RUSER(*)
  EXTERNAL        FCN
```

3 Description

The system of equations is defined as:

$$f_i(x_1, x_2, \dots, x_n) = 0, \quad i = 1, 2, \dots, n.$$

C05QBF is based on the MINPACK routine HYBRD1 (see Moré *et al.* (1980)). It chooses the correction at each step as a convex combination of the Newton and scaled gradient directions. The Jacobian is updated by the rank-1 method of Broyden. At the starting point, the Jacobian is approximated by forward differences, but these are not used again until the rank-1 method fails to produce satisfactory progress. For more details see Powell (1970).

4 References

Moré J J, Garbow B S and Hillstom K E (1980) User guide for MINPACK-1 *Technical Report ANL-80-74* Argonne National Laboratory

Powell M J D (1970) A hybrid method for nonlinear algebraic equations *Numerical Methods for Nonlinear Algebraic Equations* (ed P Rabinowitz) Gordon and Breach

5 Arguments

- 1: FCN – SUBROUTINE, supplied by the user. *External Procedure*
 FCN must return the values of the functions f_i at a point x .

The specification of FCN is:

```
SUBROUTINE FCN (N, X, FVEC, IUSER, RUSER, IFLAG)
  INTEGER          N, IUSER(*), IFLAG
  REAL (KIND=nag_wp) X(N), FVEC(N), RUSER(*)
```

1: N – INTEGER *Input*

On entry: n , the number of equations.

2: X(N) – REAL (KIND=nag_wp) array *Input*

On entry: the components of the point x at which the functions must be evaluated.

3:	FVEC(N) – REAL (KIND=nag_wp) array	Output
	<i>On exit:</i> the function values $f_i(x)$ (unless IFLAG is set to a negative value by FCN).	
4:	IUSER(*) – INTEGER array	User Workspace
5:	RUSER(*) – REAL (KIND=nag_wp) array	User Workspace
	FCN is called with the arguments IUSER and RUSER as supplied to C05QBF. You should use the arrays IUSER and RUSER to supply information to FCN.	
6:	IFLAG – INTEGER	Input/Output
	<i>On entry:</i> IFLAG > 0.	
	<i>On exit:</i> in general, IFLAG should not be reset by FCN. If, however, you wish to terminate execution (perhaps because some illegal point X has been reached), then IFLAG should be set to a negative integer.	

FCN must either be a module subprogram USED by, or declared as EXTERNAL in, the (sub) program from which C05QBF is called. Arguments denoted as *Input* must **not** be changed by this procedure.

- 2: N – INTEGER *Input*
On entry: n , the number of equations.
Constraint: $N > 0$.
- 3: X(N) – REAL (KIND=nag_wp) array *Input/Output*
On entry: an initial guess at the solution vector.
On exit: the final estimate of the solution vector.
- 4: FVEC(N) – REAL (KIND=nag_wp) array *Output*
On exit: the function values at the final point returned in X.
- 5: XTOL – REAL (KIND=nag_wp) *Input*
On entry: the accuracy in X to which the solution is required.
Suggested value: $\sqrt{\epsilon}$, where ϵ is the **machine precision** returned by X02AJF.
Constraint: $XTOL \geq 0.0$.
- 6: IUSER(*) – INTEGER array *User Workspace*
7: RUSER(*) – REAL (KIND=nag_wp) array *User Workspace*
IUSER and RUSER are not used by C05QBF, but are passed directly to FCN and should be used to pass information to this routine.
- 8: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by $X04AAF$).

Errors or warnings detected by the routine:

$IFAIL = 2$

There have been at least $200 \times (N + 1)$ calls to FCN . Consider restarting the calculation from the point held in X .

$IFAIL = 3$

No further improvement in the solution is possible. $XTOL$ is too small: $XTOL = \langle value \rangle$.

$IFAIL = 4$

The iteration is not making good progress. This failure exit may indicate that the system does not have a zero, or that the solution is very close to the origin (see Section 7). Otherwise, rerunning $C05QBF$ from a different starting point may avoid the region of difficulty.

$IFAIL = 5$

$IFLAG$ was set negative in FCN . $IFLAG = \langle value \rangle$.

$IFAIL = 11$

On entry, $N = \langle value \rangle$.
Constraint: $N > 0$.

$IFAIL = 12$

On entry, $XTOL = \langle value \rangle$.
Constraint: $XTOL \geq 0.0$.

$IFAIL = -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

$IFAIL = -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

If \hat{x} is the true solution, $C05QBF$ tries to ensure that

$$\|x - \hat{x}\|_2 \leq XTOL \times \|\hat{x}\|_2.$$

If this condition is satisfied with $XTOL = 10^{-k}$, then the larger components of x have k significant decimal digits. There is a danger that the smaller components of x may have large relative errors, but the fast rate of convergence of $C05QBF$ usually obviates this possibility.

If $XTOL$ is less than *machine precision* and the above test is satisfied with the *machine precision* in place of $XTOL$, then the routine exits with $IFAIL = 3$.

Note: this convergence test is based purely on relative error, and may not indicate convergence if the solution is very close to the origin.

The convergence test assumes that the functions are reasonably well behaved. If this condition is not satisfied, then C05QBF may incorrectly indicate convergence. The validity of the answer can be checked, for example, by rerunning C05QBF with a lower value for XTOL.

8 Parallelism and Performance

C05QBF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

C05QBF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

Local workspace arrays of fixed lengths are allocated internally by C05QBF. The total size of these arrays amounts to $n \times (3 \times n + 13)/2$ real elements.

The time required by C05QBF to solve a given problem depends on n , the behaviour of the functions, the accuracy requested and the starting point. The number of arithmetic operations executed by C05QBF to process each evaluation of the functions is approximately $11.5 \times n^2$. The timing of C05QBF is strongly influenced by the time spent evaluating the functions.

Ideally the problem should be scaled so that, at the solution, the function values are of comparable magnitude.

10 Example

This example determines the values x_1, \dots, x_9 which satisfy the tridiagonal equations:

$$\begin{aligned} (3 - 2x_1)x_1 - 2x_2 &= -1, \\ -x_{i-1} + (3 - 2x_i)x_i - 2x_{i+1} &= -1, \quad i = 2, 3, \dots, 8 \\ -x_8 + (3 - 2x_9)x_9 &= -1. \end{aligned}$$

10.1 Program Text

```
! C05QBF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module c05qbfe_mod

! C05QBF Example Program Module:
! Parameters and User-defined Routines

! .. Use Statements ..
Use nag_library, Only: nag_wp
! .. Implicit None Statement ..
Implicit None
! .. Accessibility Statements ..
Private
Public
! .. Parameters ..
Integer, Parameter, Public :: n = 9, nout = 6
Contains
Subroutine fcn(n,x,fvec,iuser,ruser,iflag)

! .. Scalar Arguments ..
Integer, Intent (Inout) :: iflag
```

```

        Integer, Intent (In)          :: n
!      .. Array Arguments ..
        Real (Kind=nag_wp), Intent (Out) :: fvec(n)
        Real (Kind=nag_wp), Intent (Inout) :: ruser(*)
        Real (Kind=nag_wp), Intent (In) :: x(n)
        Integer, Intent (Inout)      :: iuser(*)
!      .. Executable Statements ..
        fvec(1:n) = (3.0_nag_wp-2.0_nag_wp*x(1:n))*x(1:n) + 1.0_nag_wp
        fvec(2:n) = fvec(2:n) - x(1:(n-1))
        fvec(1:(n-1)) = fvec(1:(n-1)) - 2.0_nag_wp*x(2:n)
!      Set iflag negative to terminate execution for any reason.
        iflag = 0
        Return
    End Subroutine fcn
End Module c05qbfe_mod
Program c05qbfe

!      C05QBF Example Main Program

!      .. Use Statements ..
    Use nag_library, Only: c05qbf, dnorm2, nag_wp, x02ajf
    Use c05qbfe_mod, Only: fcn, n, nout
!      .. Implicit None Statement ..
    Implicit None
!      .. Local Scalars ..
    Real (Kind=nag_wp)          :: fnorm, xtol
    Integer                    :: i, ifail
!      .. Local Arrays ..
    Real (Kind=nag_wp), Allocatable :: fvec(:), x(:)
    Real (Kind=nag_wp)          :: ruser(1)
    Integer                    :: iuser(1)
!      .. Intrinsic Procedures ..
    Intrinsic                  :: sqrt
!      .. Executable Statements ..
    Write (nout,*) 'C05QBF Example Program Results'

    Allocate (fvec(n),x(n))

!      The following starting values provide a rough solution.

    x(1:n) = -1.0E0_nag_wp
    xtol = sqrt(x02ajf())

    ifail = -1
    Call c05qbf(fcn,n,x,fvec,xtol,iuser,ruser,ifail)

    If (ifail==0 .Or. ifail==2 .Or. ifail==3 .Or. ifail==4) Then
        If (ifail==0) Then
!          The NAG name equivalent of dnorm2 is f06ejf
            fnorm = dnorm2(n,fvec,1)
            Write (nout,*)
            Write (nout,99999) 'Final 2-norm of the residuals =', fnorm
            Write (nout,*)
            Write (nout,*) 'Final approximate solution'
        Else
            Write (nout,*)
            Write (nout,*) 'Approximate solution'
        End If
        Write (nout,*)
        Write (nout,99998)(x(i),i=1,n)
    End If

99999 Format (1X,A,E12.4)
99998 Format (1X,3F12.4)
    End Program c05qbfe

```

10.2 Program Data

None.

10.3 Program Results

C05QBF Example Program Results

Final 2-norm of the residuals = 0.1193E-07

Final approximate solution

-0.5707	-0.6816	-0.7017
-0.7042	-0.7014	-0.6919
-0.6658	-0.5960	-0.4164
